# Lezione 5 (parte prima)

Enrico Bertolazzi

```
> # definisco la generica formula
  # di quadratura di Gauss per 2 punti.
  # le incognite sono w1, w2 e x1, x2.

  GL2 := (f) -> w1 * f(x1) + w2 * f(x2) ;
```

$$GL2 := f \to w1\, f(x1) + w2\, f(x2) \tag{1}$$

```
> # Definisco i polinomi sui quali voglio
  # integri esattamente!

  p0 := x -> 1   ;
  p1 := x -> x   ;
  p2 := x -> x^2 ;
  p3 := x -> x^3 ;
```

$$p0 := 1 \tag{2}$$

$$p1 := x \to x$$

$$p2 := x \to x^2$$

$$p3 := x \to x^3$$

```
> # impongo le condizioni di integrazione esatta.

  eq0 := GL2(p0) = int(p0(x),x=-1..1) ;
  eq1 := GL2(p1) = int(p1(x),x=-1..1) ;
  eq2 := GL2(p2) = int(p2(x),x=-1..1) ;
  eq3 := GL2(p3) = int(p3(x),x=-1..1) ;
```

$$eq0 := w1 + w2 = 2 \tag{3}$$

$$eq1 := w1\, x1 + w2\, x2 = 0$$

$$eq2 := w1\, x1^2 + w2\, x2^2 = \frac{2}{3}$$

$$eq3 := w1\, x1^3 + w2\, x2^3 = 0$$

```
> # risolvo il sistema non lineare delle condizioni
  # di integrazione.

  res := solve({eq0,eq1,eq2,eq3},{x1,x2,w1,w2}) ;
```

$$res := \left\{w2 = 1, x1 = RootOf\left(3\,\_Z^2 - 1\right), x2 = -RootOf\left(3\,\_Z^2 - 1\right), w1 = 1\right\} \tag{4}$$

```
> evalf(res) ;
```

$$\{x1 = -0.5773502692, w2 = 1., w1 = 1., x2 = 0.5773502692\} \tag{5}$$

```
> # definisco la generica formula
  # di quadratura di Gauss per 3 punti.
  # le incognite sono w1, w2, w3 e x1, x2, x3.

  GL3 := (f) -> w1*f(x1) + w2*f(x2) + w3*f(x3) ;
```

$$GL3 := f \to w1\, f(x1) + w2\, f(x2) + w3\, f(x3) \tag{6}$$

```
> # Aggiungo i polinomi sui quali voglio
  # integri esattamente!

  p4 := x -> x^4 ;
  p5 := x -> x^5 ;
```

$$p4 := x \to x^4 \tag{7}$$

$$p5 := x \to x^5$$

```
> # impongo le condizioni di integrazione esatta.

  eq0 := GL3(p0) = int(p0(x),x=-1..1) :
  eq1 := GL3(p1) = int(p1(x),x=-1..1) :
  eq2 := GL3(p2) = int(p2(x),x=-1..1) :
  eq3 := GL3(p3) = int(p3(x),x=-1..1) :
  eq4 := GL3(p4) = int(p4(x),x=-1..1) :
  eq5 := GL3(p5) = int(p5(x),x=-1..1) :
```

```
> # risolvo il sistema non lineare delle condizioni
  # di integrazione.

  res := solve({eq0,eq1,eq2,eq3,eq4,eq5},
               {x1,x2,x3,w1,w2,w3});
```

$$res := \left\{w3 = \frac{8}{9}, w2 = \frac{5}{9}, x2 = -RootOf\left(-3 + 5\_Z^2\right), w1 = \frac{5}{9}, x3 = 0, x1 = RootOf\left(-3 + 5\_Z^2\right)\right\}, \left\{w2 = \frac{5}{9}, w3 = \frac{5}{9}, x2 = -RootOf\left(-3 + 5\_Z^2\right), w1 = \frac{8}{9}, x1 = 0, x3 = RootOf\left(-3 + 5\_Z^2\right)\right\}, \left\{w3 = \frac{5}{9}, w2 = \frac{8}{9}, x1 = -RootOf\left(-3 + 5\_Z^2\right), x2 = 0, w1 = \frac{5}{9}, x3 = RootOf\left(-3 + 5\_Z^2\right)\right\} \tag{8}$$

```
> # trovo 3 soluzioni equivalenti, scelgo la prima!

  res[1] ;
```

$$\left\{w3 = \frac{8}{9}, w2 = \frac{5}{9}, x2 = -RootOf\left(-3 + 5\_Z^2\right), w1 = \frac{5}{9}, x3 = 0, x1 = RootOf\left(-3 + 5\_Z^2\right)\right\} \tag{9}$$

```
> evalf(res[1]) ;
```

(10)

$\{x2 = 0.7745966692, w3 = 0.8888888889, w2 = 0.5555555556, x3 = 0., x1 = -0.7745966692, w1 = 0.5555555556\}$ **(10)**

```
> {x1 = -.7745966692, x2 = .7745966692, w2 = .5555555556, w3 =
  .8888888889, w1 = .5555555556, x3 = 0.};
```

$\{x2 = 0.7745966692, w3 = 0.8888888889, w2 = 0.5555555556, x3 = 0., x1 = -0.7745966692, w1 = 0.5555555556\}$ **(11)**